




Programiranje kroz aplikacije

Rad sa stringovima

Rad sa vremenom i datumima

Osnovno o stringovima

- String kao tip podatka služi za smještanje teksta i rad sa tekstom.
- Postoje stringovi promjenljive i fiksne dužine.
- Stringovi promjenljive dužine mogu sadržati i do dvije milijarde (tačnije, 2^{31}) karaktera; zauzimaju 10 bajtova plus memoriju koja je potrebna za smještanje stringa. Deklarišu se na sljedeći način:

Dim S as String

- Stringovi fiksne dužine mogu sadržati od 1 do 64 000 karaktera (tačnije, 2^{16}) i oni zauzimaju samo memoriju potrebnu za smještanje stringa. Ne deklarišu se kao nizovi, već na sljedeći način:

Dim S as String * 50


- Dodjela vrijednosti **String** promjenljivoj se vrši sa:

S = "Divan dan 26.10.2021"

Nadovezivanje stringova. vbCrLf i vbTab

- Za nadovezivanje stringova se mogu koristiti karakteri **+** i **&**.
`S = "Divan " + "dan "`
`S = S & "26.10.2011"`
- Preporučuje se korišćenje **&** umesto **+**, jer je **+** i operator sabiranja.
- Pri formiranju/nadovezivanju stringova mogu se koristiti specijalni karakteri **vbCrLf** (prelazak u novi red) i **vbTab** (tab karakter).
- Na primjer:

```
S = "aaa" & vbTab & "a" & vbCrLf  
S = S & "bb" & vbTab & "b" & vbCrLf  
S = S & "c" & vbTab & "c"
```



Funkcija Mid

- Karakterima stringa se ne pristupa kao članovima niza (C, MATLAB).
- Za pristupanje pojedinim karakterima i podstringovima stringa, koristi se funkcija **Mid**.

- **Mid(S, start, N)** vraća string koji se sastoji od **N** karaktera stringa **S** počev od karaktera čija je pozicija **start**. Argument **N** je opcion i ako se ne navede, ide se do kraja stringa. Na primjer:

Mid("Program", 2, 4) vraća "rogr"

Mid("Program", 2) vraća "rogram"

- Izmjena podstringova datog stringa se takođe vrši pomoću **Mid**. Na primjer:

Mid(S, 1, 3) = "123"

menja prva tri karaktera stringa **S** podstringom "123".

Poređenje stringova

- Za poređenje stringova može poslužiti funkcija **StrComp**.
- **StrComp(s1, s2, nacin)** poredi stringove **s1** i **s2** i vraća **0** ako su jednaki, **-1** ako je **s1 < s2**, i **1** ako je **s2 < s1**.
- Argument **nacin** predstavlja način poređenja stringova i može biti **binaran** (**nacin = 0**), kad se pravi razlika između velikih i malih slova, i **tekstualan** (**nacin = 1**), kad se ne pravi razlika. Ako se **nacin** ne navede, podrazumijevano je binarno poređenje. Na primjer:
StrComp("VBA", "vba", 0) vraća **-1**
StrComp("VBA", "vba", 1) vraća **0**
- Stringovi se mogu porediti i operatorom ispitivanja jednakosti **=**, koji vrši binarno poređenje (podrazumijevani način poređenja).
- Podrazumijevani način poređenja se definiše navođenjem **Option Compare Text** ili **Option Compare Binary** na vrhu modula.

Funkcije Len, Str i Val

- Dužina stringa se dobija pomoću funkcije **Len**, tj. **Len(S)** vraća dužinu (broj karaktera) stringa **S**.
- Funkcija **Str** konvertuje broj **X**, argument funkcije, u odgovarajući string. Na primjer:
Str(-412.56) vraća string **"-412.56"**
- Funkcija **Val** vraća broj sadržan u stringu **S**, argumentu funkcije. Čitanje broja započinje s lijeva i završava se kada se naiđe na prvi nenumerički karakter. **Val** prepoznaje tačku kao decimalni separator, ali ne i zarez, ignoriše tabove i blanko znake. Na primjer:
Val("301*44") vraća broj **301**
Val("301.1*44") vraća broj **301.1**
Val("301..1*44") vraća broj **301**

Funkcije Left i Right

- Funkcija **Left(S, N)** vraća string koji se sastoji od prvih **N** karaktera stringa **S**. Na primjer:
Left("Dobar dan",4) vraća string **"Doba"**
- **Left(S, N)** vraća isto što i **Mid(S, 1, N)**.
- Funkcija **Right(S, N)** vraća string koji se sastoji od posljednjih **N** karaktera stringa **S**. Na primjer:
Right("Dobar dan",3) vraća string **"dan"**
- **Right(S, N)** vraća isto što i **Mid(S, Len(S)-N+1)**.
- Za razliku of **Mid**, funkcije **Left** i **Right** se **ne mogu koristiti za upis**, tj. naredbe tipa
Right("Dobar dan",1) = "#"
nisu dozvoljene!

Funkcije Replace, InStr i InStrRev

- Funkcija **Replace(str, s1, s2)** vraća string **str** kod koga je svaki podstring **s1** zamijenjen podstringom **s2**. Na primjer:
Replace("Banana", "na", "tu") vraća string **"Batutu"**
- Funkcija **Replace** ima dodatne opcione argumente koji definišu poziciju u stringu **str** od koje se vrši pretraga, broj zamjena koje treba izvršiti i način poređenja stringova. Pogledati Help.
- Funkcija **InStr(s1, s2)** vraća poziciju prve pojave stringa **s2** u stringu **s1**. Ukoliko ga nema, funkcija vraća **0**. Na primjer:
InStr("Banana", "ana") vraća broj **2**
- Funkcija **InStrRev(s1, s2)** vraća poziciju posljednje pojave stringa **s2** u stringu **s1**. Ukoliko ga nema, funkcija vraća **0**. Na primjer:
InStrRev("Banana", "ana") vraća broj **4**

Funkcije LCase i UCase

- Funkcija **LCase(str)** vraća string kod koga su velika slova konvertovana u mala, a ostali karakteri su neizmijenjeni. Na primjer:
LCase("#A1G2") vraća string **"#a1g2"**
- Funkcija **UCase(str)** vraća string kod koga su mala slova konvertovana u velika, a ostali karakteri su neizmijenjeni. Na primjer:
UCase("?b1c2") vraća string **"?B1C2"**

Funkcije LTrim, RTrim i Trim

- Funkcija **LTrim(str)** eliminiše spejsove kojima počinje string **str**. Na primjer:
LTrim(" Makro") vraća string **"Makro"**
- Funkcija **RTrim(str)** eliminiše spejsove kojima se završava string **str**. Na primjer:
RTrim("Makro ") vraća string **"Makro"**
- Funkcija **Trim(str)** eliminiše spejsove kojima počinje i završava se string **str**. Na primjer:
Trim(" Makro ") vraća string **"Makro"**

Funkcije Split i Join

- Funkcija **Split(str, kar)** dijeli string **str** na niz stringova, pri čemu se dijeljenje vrši na karakteru **kar**. **Funkcija vraća niz stringova.**
- Funkcija **Join(niz, kar)** vraća string dobijen spajanjem svih stringova iz niza **niz**, pri čemu se stringovi razdvajaju karakterom **kar**.
- U obje funkcije, ako se **kar** izostavi, podrazumijeva se da je spejs.

```
Sub SplitJoin(S As String, K As String)
    Dim Niz() As String, I As Integer, novi As String
    Niz = Split(S, K)
    Debug.Print "Nakon funkcije Split"
    For I = LBound(Niz) To UBound(Niz)
        Debug.Print Niz(I)
    Next
    Debug.Print "Nakon funkcije Join"
    novi = Join(Niz, "%")
    Debug.Print novi
End Sub
```

SplitJoin "Danas je fin dan", " "



Nakon funkcije Split
Danas
je
fin
dan
Nakon funkcije Join
Danas%je%fin%dan

Rad sa kôdovima karaktera

- Funkcija **Asc(str)** vraća ANSI kôd prvog karaktera u stringu **str**. Na primjer:
Asc("A") vraća broj **65**
- Funkcija **AscW(str)** vraća Unicode kôd prvog karaktera u stringu **str**. Na primjer:
AscW("Žarko") vraća broj **381**
- Funkcija **Chr(kodkar)** vraća karakter koji odgovara ANSI kôdu **kodkar**. Na primjer:
Chr(65) vraća karakter **"A"**
- Funkcija **ChrW(kodkar)** vraća karakter koji odgovara Unicode kôdu **kodkar**. Na primjer:
ChrW(382) vraća karakter **"ž"**

Rad sa slovima š, đ, č, ć i ž

- Za rad sa našim slovima koristimo funkciju **ChrW**. Jedino što treba da znamo su Unicode kodovi naših slova (tabela ispod).

Slovo	Š	Đ	Č	Ć	Ž	Ś	š	đ	č	ć	ž	ś
Kôd	352	272	268	262	381	346	353	273	269	263	382	347

- Na primjer, string "**Šećer**" bi se definisao na sljedeći način:
S = ChrW(352) + "e" + ChrW(263) + "er"
- Na ovaj način ne možemo prikazati ova slova u korisničkim formama. Da bi to uradili, sistemski moramo podesiti opciju **Language for non-Unicode programs** na Serbian (Latin). Ova opcija se nalazi na tabu **Administrative** prozora **Region and Language** kod **Windows 7**, odnosno prozora **Region** kod **Windows 8 i 10**.
- Sa ovom opcijom omogućavamo unos naših slova u VBE prozor.
- Povedite računa o VBE fontu (Tools/Options/Editor Format/Font). Na primjer, **Courier (Western)** podržava naše karaktere.

Like operator

- **Like** operator omogućava moćno i elegantno poređenje stringova. Primjenjuje se u sljedećem obliku:

string Like obrazac

gde **obrazac** predstavlja string koji može sadržati tri džoker znaka:

- ?** mijenja bilo koji karakter
- *** mijenja proizvoljan broj karaktera
- #** mijenja bilo koju cifru

- Nekoliko primjera:

"Konik" Like "Kon?k"	vraća True
"Konik" Like "Ko?k"	vraća False
"Konik" Like "K*"	vraća True
"2023" Like "####"	vraća True
"Konik 2023" Like "K*##23"	vraća True

Like operator (nastavak)

- **Like** dozvoljava i pretragu samo određenih karaktera, koji se navode u srednjim zagradama []. Na primjer, izraz "Konik" Like "Kon[aei]k" vraća **True**, jer se na četvrtoj poziciji traže samo karakteri **a**, **e**, **i**.
- Opseg karaktera se takođe navodi u [], koristeći karakter -. Na primjer:
 - [a-z] mijenja bilo koje malo slovo
 - [A-Z] mijenja bilo koje veliko slovo
 - [A-D] mijenja velika slova **A**, **B**, **C** i **D**.
- Prilikom korišćenja opsega, opseg se mora navesti u rastućem redosljedu. Na primjer, zapis [Z-A] bi doveo do greške u izvršavanju.
- Ako ispred opsega stavimo znak uzvika, onda provjeravamo da li karakter ne pripada datom opsegu. Na primjer, sa [!a-z] ispitujemo da li dati karakter nije malo slovo.

Like operator (nastavak)

- Nekoliko primjera:

"H" Like "[A-Z]" vraća True

"H" Like "[!A-Z]" vraća False

"H5N1" Like "H#[!C-E]#" vraća True

- Ako na vrhu modula stoji **Option Compare Text**, poređenje sa [A-Z] i [a-z] daje isti rezultat.

- Provjera da li tekući karakter pripada jednom od više opsega se vrši nadovezivanjem opsega u []. Na primjer, proverda li je karakter slovo, veliko ili malo, se može vršiti na sljedeći način:

"H" Like "[a-zA-Z]" vraća True

- Iako na prvi pogled deluje logično da se opsezi razdvoje zarezima, to ne treba raditi jer bi se onda i zarez uzimao u obzir prilikom pretrage.

Like operator – Prikaz specijalnih karaktera

- Vidjeli smo da se kao specijalni karakteri u stringu obrascu koriste **[**, **]**, *****, **#** i **?**. Postavlja se pitanje - Kako koristiti ove karaktere kao ostale karaktere?
- Odgovor – Potrebno ih je smestiti unutar zagrada **[]**.
- Primjer:
"H#?1" Like "H[#][?]" vraća True
- U prethodnom primjeru, **[#]** u "H[#][?]" se odnosi na karakter **#**, a **[?]** na jednu cifru. Slično, **[?]** se odnosi na karakter **?**.

Rad sa vremenom i datumima

- VBA posjeduje tip podatka **Date** koji radi sa vremenom i datumima. VBA može da radi sa datumima iz opsega 1.1.100–31.12.9999. i vremenom od 0:00:00 do 23:59:59.
- Dodjeljivanje vrijednosti **Date** promjenljivoj se vrši pomoću karaktera taraba (#) kojim počinje i završava se vrijednost. Na primjer, datum 26.10.2021. bi se u **Date** promjenljivu **Datum** unio na bilo koji od sljedećih načina:

Datum = #October 26, 2021#

Datum = #26/10/2021#

- Vrijeme se takođe unosi pomoću karaktera #. Na primjer, **Date** promjenljivoj **Vrijeme** se dodjela vrši na sljedeći način:

Vrijeme = #18:15:22#

- U Code prozoru, VBA konvertuje unešeni podatak u format koji je sistemski podešen.

Funkcije Date, Time i Now

- Funkcija **Date** vraća tekući datum.
- Smislene su operacije sabiranja i oduzimanja sa datumima. Na primjer, datum koji je za 123 dana ispred današnjeg se dobija sa **Date + 123**
- Funkcija **Time** vraća tekuće vrijeme.
- Funkcija **Now** vraća tekući datum i vrijeme.

Funkcija DateDiff

- Funkcija **DateDiff** vraća broj proteklih vremenskih jedinica između dva datuma. Sintaksa je:

DateDiff(VremJed, Datum1, Datum2)

gdje argument **VremJed** predstavlja vremensku jedinicu u kojoj mjerimo razliku ("**yyyy**" - godine, "**m**" - mjeseci, "**w**" - nedjelje, "**d**" - dani, "**h**" - sati, "**n**" - minuti, "**s**" - sekunde), dok su **Datum1** i **Datum2** datumi između kojih se mjeri razlika.

- Na primjer, broj proteklih dana između 1. septembra 1939. i danas je **DateDiff("d", #1 September 1939#, Date)**

Funkcija Format

- Funkcija **Format** je vrlo pogodan alat pri formatiranju numeričkih vrijednosti, vremena i datuma. Mi ćemo koristiti njen skraćeni oblik:

Format(izraz, format)

gdje se **izraz** formatira u skladu sa formatom **format**.

- Funkcija vraća string koji predstavlja formatirani izraz.
- Pri formatiranju izraza, možemo koristiti *predefinisane* i *korisničke* formate. Podrazumijevani format zavisi od sistemskih podešavanja (Control Panel/Regional Options).

Predefinisani numerički formati

Format	Opis
General Number	Podrazumijevanani format. Primjer: <code>Format(12345,"General Number")</code> vraća "12345".
Currency	Prikazuje se separator hiljada, oznaka valute i dva decimalna mjesta. Oznaka valute zavisi od sistemskog podešavanja. Primjer: <code>Format(12345,"Currency")</code> vraća "12.345,00 €".
Fixed	Prikazuje se najmanje jedna cifra lijevo od decimalnog zareza i najmanje dvij cifre desno od decimalnog zareza. Primjer: <code>Format(1222.345,"Fixed")</code> vraća "1222,35".
Standard	Prikazuje se separator hiljada, najmanje jedna cifra lijevo od decimalnog zareza i najmanje dvije cifre desno od zareza. Primjer: <code>Format(1222.345,"Standard")</code> vraća "1.222,35".
Percent	Prikazuje se broj pomnožen sa 100, dvije cifre desno od decimalnog zareza i karakter procenat desno od zareza. Primjer: <code>Format(0.12345,"Percent")</code> vraća "12,35%".
Scientific	Standardni scientific zapis. Cifra najveće težine je prikazana lijevo od decimalnog zareza, i prikazuje se 2 do 30 decimalnih mjesta praćeno sa "E" i eksponentom. Primjer: <code>Format(12300,"Scientific")</code> vraća "1,23E+04".

Predefinisani formati za datum i vrijeme

Format	Opis
General Date	Podrazumijevani format.
Long Date	Prikazuje se datum u skladu sa sistemski definisanim long date formatom (na primjer, 3. decembar 2017).
Medium Date	Prikazuje se datum u skladu sa sistemski definisanim medium date formatom (na primjer, 03-dec-17).
Short Date	Prikazuje se datum u skladu sa sistemski definisanim short date formatom (na primjer, 3.12.2017).
Long Time	Prikazuje se vrijeme u skladu sa sistemski definisanim long time formatom (na primjer, 12:32:45).
Medium Time	Prikazuje se vrijeme u skladu sa sistemski definisanim medium time formatom (na primjer, 12:32).
Short Time	Prikazuje se vrijeme u skladu sa sistemski definisanim short time formatom (na primjer, 12:32).

Korisnički numerički formati

- Specijalni simboli koji se koriste pri definisanju korisničkih numeričkih formata su dati u tabeli ispod.

Simbol	Opis
#	Jedna cifra. Ništa se ne prikazuje ako nema unosa.
0	Jedna cifra. Prikazuje nulu ako nema unosa.
.	Pozicija decimalne tačke.
,	Pozicija separatora hiljada. Označava poziciju samo prve hiljade.
%	Množi broj sa 100 (samo za prikaz) i dodaje karakter %.
E+ e+ E- e-	Prikaz brojeva u scientific formatu. E- i e- smještaju znak - u eksponent; E+ i e+ smještaju znak + u eksponent.
\$ () - + <space>	Prikazuje date karaktere.
\	Umetanje karaktera koji dolazi posle \.
"text"	Umetanje teksta pod znacima navoda.

Korisnički numerički formati (primjeri)

`Format(6353.4, "##,##0.00")` vraća "6,353.40"

`Format(334.9, "###0.00 kg")` vraća "334.90 kg"

`Format(4, "0.00%")` vraća "400.00%"

`Format(6353.4, "00.00000E+")` vraća "63.53400E+2"

`Format(6353.4, "#.#E+")` vraća "6.4E+3"

`Format(1234567, "00-00-000")` vraća "12-34-567"

Primjer 1

- Napisati funkciju **Obrnut** koja za argument ima string **S** i vraća string koji ima obrnut redosljed karaktera u odnosu na **S**.

```
Function Obrnut(S As String) As String
    Dim I As Integer
    Obrnut = ""
    For I = 1 To Len(S)
        Obrnut = Mid(S, I, 1) & Obrnut
    Next
End Function
```

Primjer 2

- Napisati funkciju **BezSlova** koja za argument ima string **S** i koja vraća string koji se dobija kada se iz ulaznog stringa izbace slova.

```
Function BezSlova(S As String) As String
    Dim I As Integer
    BezSlova = ""
    For I = 1 To Len(S)
        If Mid(S, I, 1) Like "[!A-Za-z]" Then
            BezSlova = BezSlova & Mid(S, I, 1)
        End If
    Next I
End Function
```

Primjer 3

- Napisati funkciju **ObrnilmePrez** koja za argument ima string koji predstavlja ime i prezime studenta (razdvojeni spejsom) i koja treba da obrne redosljed imena i prezimena (prvo prezime, pa ime).

```
Function ObrnilmePrez(S As String) As String
    Dim K As Integer
    K = InStr(S, " ")
    ObrnilmePrez = Mid(S, K + 1, Len(S) - K) & _
                    " " & Mid(S, 1, K - 1)
End Function
```

Primjer 4

- Napisati funkciju **DecUBin** koja za argument ima prirodan broj **N** i koja vraća string koji predstavlja binarni zapis broja **N**.

```
Function DecUBin(N As Integer) As String
    Dim Cifra as Integer
    DecUBin = ""
    Do Until N = 0
        Cifra = N Mod 2
        DecUBin = Cifra & DecUBin
        N = N \ 2
    Loop
End Function
```